Program-54

(form üzerinde rastgele bir noktadan başlayarak yazma)

' Printing to the form

Option Explicit

Private Sub cmdPrint_Click()

   Call Randomize

   ' Randomly pick values to draw at

   CurrentX = Rnd() * Width

   CurrentY = Rnd() * Height

   ' Print at (CurrentX, CurrentY)

   Print "VB6 How To Program"

End Sub

Program-55

(ekranda rastgele bir noktaya rastgele dikdörtgen veya çizgi çizme)

Option Explicit

Private Sub cmdPrint_Click()

   Dim x1 As Single, y1 As Single, x2 As Single, y2 As Single

   Call Randomize

   ' Randomly pick values to draw at

   x1 = Rnd() * Width

   y1 = Rnd() * Height

   x2 = Rnd() * Width

   y2 = Rnd() * Height

   ' Randomly decide which shape

   If (Rnd() < 0.5) Then

      Line (x1, y1)-(x2, y2)             ' Line

   Else

      Line (x1, y1)-(x2, y2), vbRed, B   ' Rectangle

   End If

End Sub
Program 56

(ekrana rastgele daire çizme)

Option Explicit

Private Sub cmdPrint_Click()

   Dim x As Single, y As Single

   Dim r As Single

   Call Randomize

   ' Randomly pick values to draw at

   x = Rnd() * 5 '* Width * 5

   y = Rnd() * 5 '* Height * 5

   r = Rnd() * 2.5

   Scale (-1, 5)-(5, -5)

   ' Randomly decide which shape

   Select Case Int(Rnd() * 5)

      Case 0   ' Circle

         Circle (x, y), r

      Case 1   ' Ellipse

         Circle (x, y), r, , , , 1.5

      Case 2   ' Ellipse

         Circle (x, y), r, , , , 0.5

      Case 3   ' Sector

         Circle (x, y), r, vbRed, -2 * Atn(1), -4 * Atn(1)

      Case 4   ' Arc

         Circle (x, y), r, vbRed, 2 * Atn(1), 4 * Atn(1)

   End Select

End Sub

Program 57

(PSET metodu için demonstrasyon)

' Demonstrating method PSet

Option Explicit

Private Sub Form_Click()

   Call DrawShape

End Sub

Private Sub DrawShape()

   Dim x As Single, y As Single

   Dim totalRadians As Single, r As Single

   Dim a As Single, theta As Single

   Call Randomize

   Scale (3, -3)-(-3, 3)         ' Change scale

   totalRadians = 8 * Atn(1)     ' Circle in Radians

   a = 3 * Rnd()  ' Offset used in equation

   For theta = 0.001 To totalRadians Step 0.01

      r = 2 * a / Sin(2 * theta)

      x = r * Cos(theta)                ' y coordinate

      y = r * Sin(theta)                ' x coordinate

      PSet (x, y)                       ' Turn pixel on

   Next theta

End Sub

Program 58

(çizim özelliklerine ilişkin program)

' DrawMode, DrawStyle, and FillStyle properties

Option Explicit

Private Sub Form_Load()

   picPicture.ForeColor = vbRed   ' Set drawing color

End Sub

Private Sub chkAutoRedraw_Click()

   ' Update picPicture's AutoRedraw

   picPicture.AutoRedraw = chkAutoRedraw.Value

End Sub

Private Sub cmdClear_Click()

   ' Enable AutoRedraw so the PictureBox can

   ' be cleared with a call to Cls

   picPicture.AutoRedraw = True

   Call picPicture.Cls

   ' Uncheck CheckBox and set AutoRedraw to

   ' False.

   chkAutoRedraw.Value = vbUnchecked

   picPicture.AutoRedraw = False

End Sub

Private Sub cmdDraw_Click()

   Dim v As Integer

   Call Randomize

   v = txtDrawMode.Text

      If v >= vbBlackness And v <= vbWhiteness Then

      picPicture.DrawMode = v

   Else

      picPicture.DrawMode = vbCopyPen

      txtDrawMode.Text = vbCopyPen

   End If

   v = txtDrawStyle.Text

   If v >= vbSolid And v <= vbInsideSolid Then

      picPicture.DrawStyle = v

   Else

      picPicture.DrawStyle = vbSolid

      txtDrawStyle.Text = vbSolid

   End If

   v = txtFillStyle.Text

   If v >= vbFSSolid And v <= vbDiagonalCross Then

      picPicture.FillStyle = v

   Else  ' Out of range reset to defaults

      picPicture.FillStyle = vbFSTransparent

      txtFillStyle.Text = vbFSTransparent

   End If

   ' Draw the circle

   picPicture.Circle (picPicture.Width / 2, _

                      picPicture.Height / 2), _

                      Int(2000 * Rnd())

End Sub

Program-59

(Çizgi “Line” kontrolleri)

' Using the Line control

Option Explicit

Private Sub cmdButton_Click()

   Dim v As Integer

   v = txtStyle.Text

   If v >= 0 And v <= 6 Then

      linLine.BorderStyle = v

   Else

      linLine.BorderStyle = 1

      txtStyle.Text = 1

   End If

   v = txtWidth.Text

   If v >= 1 And v <= 100 Then

      ' If v is greater than 1 some BorderStyle

      ' properties are ignored

      linLine.BorderWidth = v

   Else

      linLine.BorderWidth = 1

      txtWidth.Text = 1

   End If

End Sub

Program-60

(farklı renklerde spiraller çizme)

' Demonstrating function RGB

Option Explicit

Private Sub DrawShape()

   Dim x As Single, y As Single

   Dim totalRadians As Single, r As Single

   Dim a As Single, theta As Single

   Dim x1 As Single, y1 As Single

   Call Randomize

   Scale (3, -3)-(-3, 3)         ' Change scale

   totalRadians = 8 * Atn(1)     ' Circle in Radians

   ForeColor = RGB(Rnd() * 256, Rnd() * 256, Rnd() * 256)

   a = 3 * Rnd()  ' Offset used in equation

   For theta = 0.001 To totalRadians Step 0.01

      r = Sqr(a ^ 2 / theta)

      x = r * Cos(theta)     ' y coordinate

      y = r * Sin(theta)     ' x coordinate

      x1 = -r * Cos(theta)   ' y coordinate

      y1 = -r * Sin(theta)   ' x coordinate

      PSet (x, y)            ' Turn pixel on

      PSet (x1, y1)          ' Turn pixel on

   Next theta

End Sub

Private Sub tmrTimer_Timer()

   Call DrawShape   ' Call when time interval expires

End Sub
Program-61

(QBColor fonksiyonu)

' Demonstrating function QBColor

Option Explicit

Private Sub DrawShape()

   Dim x As Single, y As Single

   Dim totalRadians As Single, r As Single

   Dim a As Single, theta As Single

   Call Randomize

   Scale (3, -3)-(-3, 3)         ' Change scale

   totalRadians = 8 * Atn(1)     ' Circle in Radians

   ForeColor = QBColor(Rnd() * 15)

   a = 3 * Rnd()  ' Offset used in equation

   For theta = 0 To totalRadians Step 0.01

      r = a * Sin(10 * theta)

      x = r * Cos(theta)      ' y coordinate

      y = r * Sin(theta)      ' x coordinate

      PSet (x, y)             ' Turn pixel on

   Next theta

End Sub

Private Sub tmrTimer_Timer()

   Call DrawShape   ' Call when time interval expires

End Sub
Program-62

(renk “color” iletişim kutusu örneği)

' Demonstrating the color Common Dialog

Option Explicit

Private Sub DrawShape()

   Dim x As Single, y As Single

   Dim totalRadians As Single, r As Single

   Dim a As Single, theta As Single

   Dim b As Single

   Call Randomize

   Scale (3, -3)-(-3, 3)         ' Change scale

   totalRadians = 8 * Atn(1)     ' Circle in Radians

   a = 3 * Rnd()  ' Offset used in equation

   b = 2 * Rnd()  ' Offset used in equation

   For theta = 0 To totalRadians Step 0.01

      r = b - a * Cos(theta)

      x = r * Cos(theta)      ' y coordinate

      y = r * Sin(theta)      ' x coordinate

      PSet (x, y)             ' Turn pixel on

   Next theta

End Sub

Private Sub tmrTimer_Timer()

   Call DrawShape   ' Call when time interval expires

End Sub

Private Sub cmdColorChoose_Click()

   dlgColor.ShowColor          ' Display color common dialog

   ForeColor = dlgColor.Color  ' Assign color to ForeColor

End Sub
Program-63

(bir resme “inverse” uygulama)

' Generating the inverse of an image

Private Sub Form_Load()

   picPicture.Picture = LoadPicture("c:\images\classhands.gif")

End Sub

Private Sub cmdInvert_Click()

   Call picPicture.PaintPicture(picPicture.Picture, 0, 0, _

                                , , , , , , vbDstInvert)

   Call SavePicture(picPicture, "c:\images\" & _

                    "classhands_inverse.gif")

End Sub
Program-64

(formu yazdırma)

' Demonstrating method PrintForm and the Printer object

Option Explicit

Private Sub cmdPrint_Click()

   Call PrintForm       ' Print the form

   ' Printing appears on a separate page

   Printer.PrintQuality = vbPRPQDraft

   Printer.Print "Some Printer property values: "

   Printer.Print "DriverName: " & Printer.DriverName

   Printer.Print "Port: " & Printer.Port

   Printer.Print "Copies: " & Printer.Copies

   Printer.EndDoc       ' Send to printer

End Sub

