
(MousePointer Özelliği)

' Changing the mouse pointer

Option Explicit     ' General declaration

Private Sub optCursor_Click(Index As Integer)

   MousePointer = Index

End Sub

(MouseEvents)

' Demonstrating mouse events

Option Explicit          ' General declaration

Private Sub Form_Load()

   Call Randomize        ' Randomize

End Sub

Private Sub cmdClear_Click()

   Call Cls     ' Clear Form

End Sub

Private Sub Form_Click()

   ' Randomly set Form ForeColor

   Select Case (1 + Int(Rnd() * 4))

      Case 1

         ForeColor = vbBlack

      Case 2

         ForeColor = vbMagenta

      Case 3

         ForeColor = vbRed

      Case 4

         ForeColor = vbBlue

   End Select

End Sub

Private Sub Form_DblClick()

   ' Randomly set Form BackColor

   Select Case (1 + Int(Rnd() * 4))

      Case 1

         BackColor = vbWhite

      Case 2

         BackColor = vbYellow

      Case 3

         BackColor = vbGreen

      Case 4

         BackColor = vbCyan

   End Select

   ' Change chkMove BackColor to Form's BackColor

   chkMove.BackColor = BackColor

End Sub

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, X As Single, _

Y As Single)

   CurrentX = X        ' Set x coordinate

   CurrentY = Y        ' Set y coordinate

   Print "MouseDown"

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

' Reverse coordinates

   CurrentX = Y

   CurrentY = X

   Print "MouseUp"

End Sub

Private Sub Form_MouseMove(Button As Integer,Shift As Integer, X As Single,Y As Single)

' If checked enable printing operations

   If chkMove.Value = 1 Then

      CurrentX = X

      CurrentY = Y

      Print "MouseMove"

   End If

End Sub

[image: image1.png]€ default

€ aow

€ crasshai

€ lbeam

€ ieon

s

€ NESW

CNS

12.2: MousePainter Propeity

€ NwSE
"3
€ upanow

€ houglass

%

%

%

rodiop

houglass
question





Option Explicit      ' General declaration

Private Sub Form_Load()

   imgImage.Picture = LoadPicture("c:\images\mouse0.gif")

End Sub

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, X As Single, _

                           Y As Single)

   Call SetPressedImage(Button)

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, _

                         X As Single, Y As Single)

   Call SetReleasedImage

End Sub

Private Sub imgImage_MouseDown(Button As Integer, Shift As Integer, _

                               X As Single, Y As Single)

   Call SetPressedImage(Button)

End Sub

Private Sub imgImage_MouseUp(Button As Integer, Shift As Integer, _

                             X As Single, Y As Single)

   Call SetReleasedImage

End Sub

Private Sub SetPressedImage(b As Integer)

   imgImage.Picture = LoadPicture("c:\images\mouse" & b & ".gif")

End Sub

Private Sub SetReleasedImage()

   imgImage.Picture = LoadPicture("c:\images\mouse0.gif")

End Sub

(Shift, Ctrl ve Alt tuşları ile mouse butonları kombinasyonu)

Option Explicit      ' General declaration

Private Sub Form_Load()

   imgImage.Picture = LoadPicture("c:\images\mouse0.gif")

End Sub

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, x As Single, _

                           Y As Single)

   Dim a As Integer, c As Integer, s As Integer

   a = Shift And vbAltMask      ' Mask Alt bit

   c = Shift And vbCtrlMask     ' Mask Ctrl bit

   s = Shift And vbShiftMask    ' Mask Shift bit

   If (a <> 0) Then     ' Alt

      lblKeys(a).BackColor = vbGreen

   End If

   If (s <> 0) Then     ' Shift

      lblKeys(s).BackColor = vbGreen

   End If

   If (c <> 0) Then     ' Ctrl

      lblKeys(c).BackColor = vbGreen

   End If

   imgImage.Picture = LoadPicture("c:\images\mouse" & Button & ".gif")

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, x As Single, Y As Single)

   ' Change Label BackColors to Form's BackColor

   lblKeys(1).BackColor = BackColor

   lblKeys(2).BackColor = BackColor

   lblKeys(4).BackColor = BackColor

   ' Load mouse image of unpressed buttons

   imgImage.Picture = LoadPicture("c:\images\mouse0.gif")

End Sub

(Manual Drag and Drop)

' Demonstrating manual drag-and-drop

Option Explicit                ' General declaration

Dim mCurrentCell As Integer    ' General declaration

Private Sub Form_Load()

   Dim x As Integer

   mCurrentCell = 2     ' Lower left corner

   For x = 1 To 64

      If x Mod 2 Then

         picSquare(x).Picture = LoadPicture("c:\images\" & "w_marble.jpg")

      Else

         picSquare(x).Picture = LoadPicture("c:\images\" & "b_marble.jpg")

      End If

   Next x

   picSquare(2).Picture = LoadPicture("c:\images\" & "b_knight.jpg")

End Sub

Private Sub picSquare_MouseDown(Index As Integer, Button As Integer, Shift As Integer, _

                                x As Single, Y As Single)

   ' If on the PictureBox displaying the image

   ' then enable dragging.

   If Index = mCurrentCell Then

      picSquare(mCurrentCell).Drag vbBeginDrag

   End If

End Sub

Private Sub picSquare_DragOver(Index As Integer, Source As Control, _

                               x As Single, Y As Single, _

                               State As Integer)

   ' Display icon while dragging over a PictureBox

   picSquare(Index).DragIcon = LoadPicture("c:\images" & _

                                           "\knight.cur")

End Sub

Private Sub picSquare_DragDrop(Index As Integer, _

                               Source As Control, _

                               x As Single, Y As Single)

   ' Draw image at new position

   If Index Mod 2 Then

      picSquare(Index).Picture = LoadPicture("c:\images" & _

                                             "\w_knight.jpg")

   Else

      picSquare(Index).Picture = LoadPicture("c:\images" & _

                                             "\b_knight.jpg")

   End If

   ' Remove last image only if the drop is at

   ' a different location.

   If mCurrentCell <> Index Then

      If Source.Index Mod 2 Then

         Source.Picture = LoadPicture("c:\images" & _

                                      "\w_marble.jpg")

      Else

         Source.Picture = LoadPicture("c:\images" & _

                                      "\b_marble.jpg")

      End If

   End If

   ' Update current image position

   mCurrentCell = Index

End Sub

(Automatic Drag and Drop)

' Demonstrating Automatic drag-and-drop

Option Explicit         ' General declaration

Private Sub Form_Load()

   Dim a As Integer

   ' Set all DragMode properties to Automatic

   For a = cmdButton.LBound To cmdButton.UBound

      cmdButton(a).DragMode = 1    ' Automatic

   Next a

End Sub

Private Sub Form_DragDrop(Source As Control, X As Single, Y As Single)

   Dim w As Integer, h As Integer

   ' Center control on mouse pointer

   w = X - Source.Width / 2

   h = Y - Source.Height / 2

   ' Move button to location where drop occurs

   Call Source.Move(w, h)

End Sub

' Demonstrating KeyDown, KeyUp, and KeyPress

Option Explicit                ' General declaration

Dim mTitleString As String     ' General declaration

Private Sub Form_Load()

   ' Store Caption value for use in KeyPress

   mTitleString = Caption & Space$(5)

End Sub

Private Sub Form_KeyDown(KeyCode As Integer, Shift As Integer)

   ' Determine which, if any, of the Shift, Ctrl,

   ' or Alt keys is pressed

   Select Case Shift

      Case vbShiftMask                          ' Shift

         ForeColor = vbYellow

      Case vbAltMask                            ' Alt

         ForeColor = vbRed

      Case vbCtrlMask                           ' Ctrl

         ForeColor = vbGreen

      Case vbShiftMask + vbAltMask              ' Shift + Alt

         ForeColor = vbBlue

      Case vbShiftMask + vbCtrlMask             ' Shift + Ctrl

         ForeColor = vbMagenta

      Case vbAltMask + vbCtrlMask               ' Alt + Ctrl

         ForeColor = vbCyan

      Case vbAltMask + vbCtrlMask + vbShiftMask ' All three

         Call Cls

    End Select

   ' Test for letter key

   If KeyCode >= vbKeyA And KeyCode <= vbKeyZ Then

      Print Chr$(KeyCode);    ' Print the character

   ElseIf KeyCode = vbKeyReturn Then   ' Return key

      Print                   ' Print on next line

   End If

End Sub

Private Sub Form_KeyPress(KeyAscii As Integer)

   ' Update title bar to display the key pressed

   Caption = mTitleString & "(" & Chr$(KeyAscii) & ")"

End Sub

Private Sub Form_KeyUp(KeyCode As Integer, Shift As Integer)

   ' When key is released, change ForeColor to black

   ForeColor = vbBlack

End Sub

(KeyPreview)

' Demonstrating the KeyPreview property.

Option Explicit            ' General declaration

Private Sub Form_Load()

   Call Randomize

   ' Allow Form to get key events first

   KeyPreview = True

End Sub

Private Sub Form_KeyPress(KeyAscii As Integer)

   ' Only allow numeric keys

   If KeyAscii >= vbKey0 And KeyAscii <= vbKey9 Then

      txtInput.Text = txtInput.Text & Chr$(KeyAscii)

   End If

End Sub

Private Sub txtInput_KeyPress(KeyAscii As Integer)

   KeyAscii = 0   ' Disable event handling

End Sub

Private Sub txtInput_KeyUp(KeyCode As Integer, Shift As Integer)

   Select Case Int(Rnd() * 3)

      Case 0

         txtInput.BackColor = vbYellow

      Case 1

         txtInput.BackColor = vbCyan

      Case 2

         txtInput.BackColor = vbRed

   End Select

End Sub

