Program-31
(Linear Search Algorithm)

(List box tıklandıktan sonraki bir ekran)

(değer yazılıp search butonu tıklandıktan sonraki ekran)

' Demonstrating a linear search

Option Explicit

Option Base 1

Dim mArray(10) As Integer

Private Sub cmdSearch_Click()

   Dim searchKey As Integer   ' Value to search for

   Dim element As Integer     ' Index of Value

   lblResult.Caption = ""

   searchKey = txtKey.Text

   ' Call LinearSearch and pass array and key

   element = LinearSearch(mArray(), searchKey)

   If element <> -1 Then

      lblResult.Caption = "Value was found."

   Else

      lblResult.Caption = "Value was not found."

   End If

End Sub

Private Sub Form_Load()

   Call lstData_Click

End Sub

Private Sub lstData_Click()

   Dim x As Integer

   Call Randomize

   Call lstData.Clear

   lblResult.Caption = ""

   ' Generate some random data

   For x = LBound(mArray) To UBound(mArray)

      mArray(x) = 1 + Int(10000 * Rnd())

      Call lstData.AddItem(mArray(x))

   Next x

End Sub

Private Sub cmdExit_Click()

   End

End Sub

' Code module modLinear.bas

Option Explicit

Function LinearSearch(a() As Integer, key As Integer) As Integer

   Dim x As Integer

   For x = LBound(a) To UBound(a)

      If a(x) = key Then

         LinearSearch = x   ' Return index

         Exit Function

      End If

   Next x

   LinearSearch = -1        ' Value not found

End Function

Program-32
(Binary Search Algorithm)

(ilk ekran)

(kullanıcı 19 girip Search butonunu tıkladıktan sonra)

(kullanıcı 8 girip Search butonunu tıkladıktan sonra)

' Demonstrating a binary search

Option Explicit

Option Base 1

Dim mArray(15) As Integer

Dim mLowBound As Integer

Dim mUpperBound As Integer

Private Sub Form_Load()

   Dim x As Integer

   mLowBound = LBound(mArray)

   mUpperBound = UBound(mArray)

   ' Generate some array data

   For x = mLowBound To mUpperBound

      mArray(x) = 2 * x

   Next x

End Sub

Private Sub cmdSearch_Click()

   Dim x As Integer

   Call Cls

   ' Print blanks so printing does not

   ' print behind Label and TextBox

   For x = 1 To 5

      Print

   Next x

   Call BinarySearch

End Sub

Private Sub BinarySearch()

   Dim middle As Integer

   Dim low As Integer, high As Integer

   Dim searchKey As Integer

   low = mLowBound

   high = mUpperBound

   Call PrintHeader

   searchKey = txtKey.Text

   Do While (low <= high)

      middle = (low + high) \ 2

      Call PrintRow(low, middle, high)

      If (searchKey = mArray(middle)) Then

         Print "Found " & searchKey & " in " _

               & "index " & middle

         Exit Sub

      ElseIf searchKey < mArray(middle) Then

         high = middle - 1

      Else

         low = middle + 1

      End If

   Loop

   Print searchKey & " not found."

End Sub

Private Sub PrintHeader()

   Dim x As Integer

   Print "Indexes:"

   For x = mLowBound To mUpperBound

      Print Format$(x, "!@@@@");

   Next x

   Print

   For x = mLowBound To 4 * mUpperBound

      Print "-";

   Next x

   Print

End Sub

Private Sub PrintRow(low As Integer, middle As Integer, high As Integer)

   Dim x As Integer

   For x = mLowBound To mUpperBound

      If (x < low Or x > high) Then

         Print Space$(4);

      ElseIf (x = middle) Then

         Print Format$(mArray(x) & "*", "!@@@@");

      Else

         Print Format$(mArray(x), "!@@@@");

      End If

   Next x

   Print

End Sub

Program-33
(Dinamik diziler)

' Demonstrating dynamic arrays

Option Explicit

Option Base 1

Dim mDynamic() As Integer

Private Sub cmdErase_Click()

   Erase mDynamic        ' Free memory

   Call lstValues.Clear

   cmdErase.Enabled = False

   chkPreserve.Value = vbUnchecked  ' Uncheck Preserve

End Sub

Private Sub cmdPrint_Click()

   Dim arrayLength As Integer, x As Integer

   Call lstValues.Clear

   arrayLength = txtInput.Text

   ' Determine state of CheckBox

   If chkPreserve.Value = vbUnchecked Then   ' Unchecked

      ReDim mDynamic(arrayLength)             ' Allocate memory

      Call InitArray                         ' Initialize to defaults

   ElseIf chkPreserve.Value = vbChecked Then ' Checked

      ' Allocate memory and preserve contents

      ReDim Preserve mDynamic(arrayLength)

   End If

   ' Display index and array contents in ListBox

   For x = LBound(mDynamic) To UBound(mDynamic)

      Call lstValues.AddItem("Index: " & x & vbTab & _

                        vbTab & "Value: " & mDynamic(x))

   Next x

   cmdErase.Enabled = True

End Sub

Private Sub InitArray()

   Dim x As Integer

   ' Generate some data

   For x = LBound(mDynamic) To UBound(mDynamic)

      mDynamic(x) = Int(51 * Rnd())

   Next x

End Sub

Program-34
(“Array” fonksiyonu)

(ilk ekran)

(Print butonu tıklandıktan sonra)

' Demonstrating function Array.

Option Explicit

Option Base 1

Private Sub cmdPrint_Click()

   Dim v As Variant, x As Integer

   v = Array(7, 5, 6, 9, 3, 0)     ' Returns Variant array

   Print "Variant array values are: ";

   For x = LBound(v) To UBound(v)

      Print Format$(v(x), "@@@");

   Next x

   Print

   v = Array("hello", "bye", "hi")  ' Return Variant array

   Print "Variant array values are: ";

   For x = LBound(v) To UBound(v)

      Print v(x) & Space$(2);

   Next x

   Print

   v = Array(1.1, 2.2, 3.3, 4.4)     ' Return Variant array

   Print "Variant array values are: ";

   For x = LBound(v) To UBound(v)

      Print v(x) & Space$(2);

   Next x

   cmdPrint.Enabled = False

End Sub
(ilk ekran)








(4 yazıp display butonuna bastıktan sonra)








(preserve tıklanıp 8 yazıp display butonuna bastıktan sonra)








(erase butonu tıklandıktan sonra)








