
' Demonstrating FileListBox, DirListBox,

' and DriveListBox controls

Option Explicit      ' General declaration

Private Sub dirDirBox_Change()

   ' Update the file path to the directory path

   filFileBox.Path = dirDirBox.Path

End Sub

Private Sub drvDriveBox_Change()

   On Error GoTo errorhandler

      ' Update the directory path to the drive

      dirDirBox.Path = drvDriveBox.Drive

      Exit Sub

errorhandler:

      Dim message As String

      ' Check for device unavailable error

      If Err.Number = 68 Then

         Dim r As Integer

         message = "Drive is not available."

         r = MsgBox(message, vbRetryCancel + vbCritical, "VBHTP: Chapter 14")

         ' Determine where control should resume

         If r = vbRetry Then

            Resume

         Else   ' Cancel was pressed.

            drvDriveBox.Drive = drvDriveBox.List(1)

            Resume Next

         End If

      Else

         Call MsgBox(Err.Description, vbOKOnly + vbExclamation)

         Resume Next

      End If

End Sub

Option Explicit                          ' General declaration

Dim mFileSysObj As New FileSystemObject  ' General declaration

Private Sub dirDirBox_Change()

   ' Update the file path to the directory path

   filFileBox.Path = dirDirBox.Path

End Sub

Private Sub drvDriveBox_Change()

   On Error GoTo errorhandler

      ' Update the directory path to the drive

      dirDirBox.Path = drvDriveBox.Drive

      Exit Sub

errorhandler:

      Dim message As String

      ' Check for device unavailable error

      If Err.Number = 68 Then

         Dim r As Integer

         message = "Drive is not available."

         r = MsgBox(message, vbRetryCancel + vbCritical, _

                    "VBHTP: Chapter 14")

         ' Determine where control should resume

         If r = vbRetry Then

            Resume

         Else   ' Cancel was pressed.

            drvDriveBox.Drive = drvDriveBox.List(1)

            Resume Next

         End If

      Else

         Call MsgBox(Err.Description, vbOKOnly + vbExclamation)

         Resume Next

      End If

End Sub

Private Sub filFileBox_Click()

   Call displayData        ' Update TextBox

End Sub

' Programmer defined

Private Sub displayData()

   txtDisplay.Text = ""

   txtDisplay.Text = "GetAbsolutePathName: " & _

                     mFileSysObj.GetAbsolutePathName( _

                     filFileBox.Path) & vbNewLine & _

                     "GetDriveName: " & _

                     mFileSysObj.GetDriveName( _

                     filFileBox.Path) & vbNewLine & _

                     "GetParentFolderName: " & _

                     mFileSysObj.GetParentFolderName( _

                     filFileBox.Path) & vbNewLine & _

                     "GetTempName: " & mFileSysObj.GetTempName

End Sub

Private Sub mnuitmCreateFolder_Click()

   Dim s As String

   ' Get the Folder name

   s = InputBox("Enter complete path and folder name:", "CREATE")

   ' Test if the Folder already exists

   If mFileSysObj.FolderExists(s) Then

      Call MsgBox("Folder already exists! Cannot create.")

      Exit Sub

   End If

   Call mFileSysObj.CreateFolder(s)   ' Create the Folder

   Call dirDirBox.Refresh             ' Repaint DirListBox

End Sub

Private Sub mnuitmDeleteFolder_Click()

  Dim s As String

   ' Get the Folder name

   s = InputBox("Enter complete path and folder name:", "DELETE")

   ' Test if the Folder already exists

   If mFileSysObj.FolderExists(s) = False Then

      Call MsgBox("Folder does not exist! Cannot delete.")

      Exit Sub

   End If

   Call mFileSysObj.DeleteFolder(s)   ' Delete the Folder

   Call dirDirBox.Refresh             ' Repaint DirListBox

End Sub

Private Sub mnuitmExit_Click()

   End   ' Terminate execution

End Sub

Option Explicit                   ' General declaration

Dim mFso As New FileSystemObject  ' General declaration

Private Sub dirDirectory_Change()

   ' Update filFile's Path when dirDirectory changes

   filFile.Path = dirDirectory.Path

End Sub

Private Sub filFile_Click()

   Dim theFile As File

   txtBox.Text = ""    ' Clear TextBox

   ' Determine which file name was selected and return

   ' its File object.

   Set theFile = mFso.GetFile(filFile.Path & "\" & _

                 filFile.List(filFile.ListIndex))

   ' Display File information in PictureBox

   txtBox.Text = "Created: " & theFile.DateCreated & _

                 vbNewLine & _

                 "Last Accessed: " & theFile.DateLastAccessed & _

                 vbNewLine & "Last Modified: " & _

                 theFile.DateLastModified & vbNewLine & _

                 "Drive: " & theFile.Drive & vbNewLine & _

                 "Size: " & theFile.Size & " bytes" & _

                 vbNewLine & _

                 "Path: " & theFile.Path & vbNewLine & _

                 "Short Name: " & theFile.ShortName

End Sub

Option Explicit                    ' General declaration

Dim mFso As New FileSystemObject   ' General declaration

Private Sub dirBox_Click()

   Dim f As Folder

   ' Get a Folder to the selected item

   Set f = mFso.GetFolder(dirBox.List(dirBox.ListIndex))

   txtDisplay.Text = ""      ' Clear TextBox

   ' Test for the root folder

   If f.IsRootFolder = False Then

      txtDisplay.Text = "Root folder: " & f.IsRootFolder _

                        & vbNewLine & _

                        "Parent Folder: " & f.ParentFolder & _

                        vbNewLine & "Size: " & f.Size

   Else

      txtDisplay.Text = "Root folder: " & f.IsRootFolder

   End If

   txtDisplay.Text = txtDisplay.Text & vbNewLine & "Type: " & _

                     f.Type & vbNewLine & _

                     "Short Path: " & f.ShortPath & _

                     vbNewLine & "Path: " & f.Path & _

                     vbNewLine & "Short Name: " & f.ShortName

End Sub

Option Explicit                   ' General declaration

Dim mFso As New FileSystemObject  ' General Declaration

Private Sub Form_Load()

   Call drvBox_Change    ' Call DriveListBox Change

End Sub

Private Sub drvBox_Change()

   Dim d As Drive

   ' Get the Drive and assign it to d

   Set d = mFso.GetDrive(drvBox.List(drvBox.ListIndex))

   ' Enable error handler

   On Error GoTo errhandler

      lblStats(0).Caption = "Letter: " & d.DriveLetter & _

                            Space$(3) & "S/N #: " _

                            & d.SerialNumber & Space$(3) & _

                            "Free space: " & d.FreeSpace / _

                            1000000 & " MB"

      lblStats(1).Caption = "Type: " & d.DriveType _

                            & Space$(3) & "File System: " _

                            & d.FileSystem & Space$(3) & _

                            "Total Size: " & d.TotalSize / _

                            1000000 & " MB"

      Exit Sub

errhandler:

      Call MsgBox(Err.Description, vbCritical, "ERROR")

      drvBox.Drive = drvBox.List(1)

      Exit Sub

End Sub

Option Explicit                          ' General declaration

Dim mFileSysObj As New FileSystemObject  ' General declaration

Dim mFile As File                        ' General declaration

Dim mTxtStream As TextStream             ' General declaration

Private Sub Form_Load()

   ' Create a text file

   Call mFileSysObj.CreateTextFile("c:\clients.dat")

   ' Once file is created, reference the file

   Set mFile = mFileSysObj.GetFile("c:\clients.dat")

   ' Open a text stream for writing to the file

   Set mTxtStream = mFile.OpenAsTextStream(ForWriting)

   ' Display path in lblFileName

   lblFileName.Caption = mFile.Path

End Sub

Private Sub cmdWrite_Click()

   ' Write the data to the file

   Call mTxtStream.WriteLine(mskAccount.Text & " " & _

                        txtName.Text & " " & txtBalance.Text)

   ' Clear MaskEdit and TextBoxes

   txtName.Text = ""

   txtBalance.Text = ""

   ' Set several properties for mskAccount

   ' using With statement

   With mskAccount

      .Text = "000"     ' Display all zeros in MaskEdit

      .SelStart = 0     ' Start highlighting at position 0

      .SelLength = 3    ' Highlight 3 characters

      .SetFocus         ' Transfer focus

   End With

End Sub

Private Sub Form_Terminate()

   Call mTxtStream.Close     ' Close the text stream

End Sub

Option Explicit                          ' General declaration

Dim mFileSysObj As New FileSystemObject  ' General declaration

Dim mFile As File                        ' General declaration

Dim mTxtStream As TextStream             ' General declaration

Private Sub Form_Load()

   ' Get the file

   Set mFile = mFileSysObj.GetFile("c:\clients.dat")

   ' Open a text stream for writing to the file

   Set mTxtStream = mFile.OpenAsTextStream(ForReading)

   ' Display path in lblFileName

   lblFileName.Caption = mFile.Path

End Sub

Private Sub cmdRead_Click()

   Dim s As String

   On Error GoTo handler   ' Set error trap

   ' Read the data

   s = mTxtStream.ReadLine

   ' Parse String s to get values for MaskEdit  and TextBoxes

   Dim mark1 As Integer, mark2 As Integer

   mark1 = 4   ' Location of first space in s

   ' Place only the String portion representing the account in the MaskEdit.

   mskAccount.Text = Trim$(Mid$(s, 1, mark1))

   ' Position to first letter of name

   mark1 = mark1 + 1

   ' Determine location of second space character and  add 1 to include the space.

   mark2 = InStr(mark1, s, " ", vbTextCompare) + 1

   ' Place only the String portion representing the name in the TextBox.

   txtName.Text = Trim$(Mid$(s, mark1, mark2 - mark1))

   ' Place the formatted dollar amount in the TextBox. 

   ' mark2 is positioned at the beginning of the amount.

   txtBalance.Text = Format$(Mid$(s, mark2, Len(s) - mark1), "Currency")

   Exit Sub

handler:

   If Err.Number = 62 Then      ' EOF error

      Call mTxtStream.Close

      cmdRead.Enabled = False

      lblFileName.Caption = ""

   Else

      Call MsgBox(Err.Description)

   End If

End Sub

Option Explicit                   ' General declaration

Dim mFso As New FileSystemObject  ' General declaration

Dim mType As Integer              ' General declaration

Const mCREDIT = 0, mDEBIT = 1, mZERO = 2  ' General declaration

Private Sub cmdButton_Click(Index As Integer)

  mType = Index   ' Assign cmdButton control array index

  Call openAndReadFile  ' Open and read file

End Sub

Private Sub openAndReadFile()

   Dim txtStream As TextStream, s As String

   Dim balance As Currency, pos As Long

   txtDisplay.Text = "Accounts:"

   ' Get "clients.dat" and open a TextStream for reading

   Set txtStream = mFso.GetFile("c:\clients.dat"). OpenAsTextStream(ForReading)

   ' Loop until end of stream is found

   Do

     s = txtStream.ReadLine   ' Read one line

     ' Find the position of the second space

     pos = InStr(InStr(1, s, " ", vbTextCompare) + 1, s, " ", vbTextCompare)

     ' Extract the String that contains the balance

     balance = Trim$(Mid$(s, pos, Len(s) - pos))

     ' Determine what if anything should be displayed

     If (mType = mCREDIT And balance < 0) Then

        txtDisplay.Text = txtDisplay.Text & vbNewLine & s

     ElseIf (mType = mDEBIT And balance > 0) Then

        txtDisplay.Text = txtDisplay.Text & vbNewLine & s

     ElseIf (mType = mZERO And balance = 0) Then

        txtDisplay.Text = txtDisplay.Text & vbNewLine & s

     End If

   Loop While (txtStream.AtEndOfStream = False)

   Call txtStream.Close     ' Close TextStream

End Sub







