Program-26
(dizi elemanlarına değer atama)

' Initializing an array

Option Explicit

Private Sub cmdPrint_Click()

   Dim s(9) As Integer

   Dim x As Integer

   Call Cls

   ' Initialize array to even elements

   For x = LBound(s) To UBound(s)

      s(x) = 2 + 2 * x

   Next x

   Print "Index" & Space$(3) & "Value"

   For x = LBound(s) To UBound(s)

      Print Space$(2) & x & Space$(7) & s(x)

   Next x

End Sub

Private Sub cmdExit_Click()

   End

End Sub

Program-27
(40 öğrenciye kafeteryadaki yemekler hakkında 5 seçenekli bir anket veriliyor (1=Çok kötü, 5=Çok iyi) Rastgele oluşturduğunuz 40 yanıtı tamsayı dizininde tutun ve anket sonuçlarını özetleyin)

Option Explicit

' Set default lower bound to 1

Option Base 1

Dim mResponses(40) As Integer

Private Sub Form_Load()

   Dim x As Integer

   ' Generate survey data

   For x = LBound(mResponses) To UBound(mResponses)

      mResponses(x) = 1 + Int(Rnd() * 10)

   Next x

End Sub

Private Sub cmdPrint_Click()

   Dim frequency(10) As Integer ' 10 Elements

   Dim x As Integer

   Call Cls

   Call Randomize

   ' Calculate results

   For x = LBound(mResponses) To UBound(mResponses)

      frequency(mResponses(x)) = frequency(mResponses(x)) + 1

   Next x

   Print "Rating" & Space$(3) & "Frequency"

   For x = LBound(frequency) To UBound(frequency)

      Print Space$(3) & x & vbTab & vbTab & frequency(x)

   Next x

   cmdPrint.Enabled = False

End Sub

Private Sub cmdExit_Click()

   End

End Sub
Program-28
(“Roll” butonuna her basıldığında 12 zar atılıyor. Gelen her yüz sıklık değeri olarak ekranda görülüyor)

(Roll butonu bir kez tıklandığında)

(Roll butonu 10 kez tıklandığında)
' Six-sided die simulation

Option Explicit

' Declare module variables

Dim mFrequency(1 To 6) As Integer

Dim mTotalRolls As Integer

Private Sub cmdRoll_Click()

   Call Randomize                          ' Seed function Rnd

   mTotalRolls = mTotalRolls + 12

   Call DisplayDie(imgDie1)

   Call DisplayDie(imgDie2)

   Call DisplayDie(imgDie3)

   Call DisplayDie(imgDie4)

   Call DisplayDie(imgDie5)

   Call DisplayDie(imgDie6)

   Call DisplayDie(imgDie7)

   Call DisplayDie(imgDie8)

   Call DisplayDie(imgDie9)

   Call DisplayDie(imgDie10)

   Call DisplayDie(imgDie11)

   Call DisplayDie(imgDie12)

   Call lstStatistics.Clear

   Call DisplayStats

End Sub

Private Sub DisplayDie(imgDie As Image)

   Dim face As Integer

   face = 1 + Int(Rnd() * 6)

   ' Replaces Select Case structure

   mFrequency(face) = mFrequency(face) + 1

   imgDie.Picture = LoadPicture("c:\images\die" & face & ".gif")

End Sub

Private Sub DisplayStats()

   Dim x As Integer

   lstStatistics.AddItem "Face" & vbTab & "Frequency" & _

                         vbTab & "Percent"

   ' Replaces multiple output statements

   For x = LBound(mFrequency) To UBound(mFrequency)

      lstStatistics.AddItem x & vbTab & mFrequency(x) & vbTab _

                            & vbTab & Format$(mFrequency(x) / _

                            mTotalRolls, "Percent")

   Next x

End Sub
Program-29
(bir diziye değer atama ve dizi elemanlarını alt programa gönderme örneği)

(ilk ekran)

(array butonu kliklendikten sonra)

(element butonu kliklendikten sonra)

' Passing an array and individual array elements to procedures

Option Explicit            ' General declaration

' Declare array with indexes between -5 and 5

Dim mArray(-5 To 5) As Integer

Private Sub cmdDisplay_Click()

   Dim x As Integer

   ' Initialize array and list boxes

   Call Initialize

   ' Pass whole array call-by-reference

   Call ModifyArray(mArray())

   ' Print modified values to list box

   Call PrintModified

End Sub

Private Sub cmdElement_Click()

   Dim x As Integer

   Call Initialize ' Initialize array

   For x = LBound(mArray) To UBound(mArray)

      ' Pass each element call-by-reference

      Call ModifyElement(mArray(x))

   Next x

   Call PrintModified

End Sub

Private Sub cmdExit_Click()

   End

End Sub

Private Sub Initialize()

   Dim x As Integer

   lstOriginal.Clear   ' Clear list boxes

   lstModified.Clear

   For x = LBound(mArray) To UBound(mArray)      ' Place value into array

      mArray(x) = x

      lstOriginal.AddItem mArray(x)

   Next x

End Sub

Private Sub PrintModified()

   Dim x As Integer

   For x = LBound(mArray) To UBound(mArray)

      lstModified.AddItem mArray(x)

   Next x

End Sub

Private Sub ModifyArray(a() As Integer)

   Dim x As Integer

   For x = LBound(a) To UBound(a)

      a(x) = a(x) * 2

   Next x

End Sub

Private Sub ModifyElement(element As Integer)

   element = element * 5

End Sub

Program-30
(dizi verilerinin sıralanması – Bubble Sort)

(ilk ekran)

(Create Data butonu kliklendikten sonra)

[image: image11.png]=lo/x|

Original values Sorted values.

Create Data Sort Exit





(Sort butonu kliklendikten sonra)

' Bubble Sort

Option Explicit

Option Base 1

Dim mArray(10) As Integer

Private Sub cmdGenerate_Click()

   Dim x As Integer

   Call Randomize          ' Randomize Rnd

   Call lstOriginal.Clear  ' Clear data

   Erase mArray            ' Clear array

   ' Generate numbers

   For x = LBound(mArray) To UBound(mArray)

      mArray(x) = 1 + Int(100 * Rnd())

      Call lstOriginal.AddItem(mArray(x))

   Next x

   Call lstSorted.Clear    ' Clear ListBox

   cmdSort.Enabled = True  ' Enable Sort button

End Sub

Private Sub cmdSort_Click()

   Dim x As Integer

   Call lstSorted.Clear    ' Clear ListBox

   Call BubbleSort(mArray) ' Sort the array

   For x = 1 To UBound(mArray)

      Call lstSorted.AddItem(mArray(x))

   Next x

   cmdSort.Enabled = False

End Sub

Private Sub cmdExit_Click()

   End

End Sub
' Module modBubble.bas

Option Explicit

Public Sub BubbleSort(theArray() As Integer)

   Dim pass As Integer, compare As Integer

   Dim hold As Integer

   For pass = 1 To (UBound(theArray) - 1)

        For compare = 1 To (UBound(theArray) - 1)

           If theArray(compare) > theArray(compare + 1) Then

            hold = theArray(compare)

            theArray(compare) = theArray(compare + 1)

            theArray(compare + 1) = hold

         End If

      Next compare

   Next pass

End Sub















